Exercise 1.9

Let X be a *Pareto* random variable (rv) with parameters $\beta > 0$ and $\theta > 0$ and therefore with probability density function (pdf) as follows:

$$f_X(x \mid \beta, \theta) = \frac{\beta \theta^\beta}{x^{\beta+1}} \mathbb{1}_{\{x \geq \theta\}}$$

Consider a portfolio of 1-year term property and casualty insurance, under the collective risk model:

$$S = Y_1 + \cdots + Y_N$$

where the *frequency* is a rv $N := \max\{n \in \{0, 1, \ldots\} : n \leq X - \theta\}$ and the conditional *severity* per claim is given by $Y \mid N = n \sim \text{Pareto}(2 + \frac{1}{n}\ , \delta)$ for $n \geq 1$. Calculate or estimate expected value, variance, median and $\text{VaR}_{0.995}$ of S and $S \mid S > 0$, with parameter values $\beta = 3, \theta = 1 = \delta$.

```
using Distributions   , Statistics   , Plots
```

```
simulateCRM (generic function with 1 method)
```

```
function simulateCRM(; β = 3.0, θ = 1.0, δ = 1.0, m = 1_000_000)
    S = zeros(m)
    N = Int.(floor.(rand(Pareto(β, θ), m) .- θ)) # m = number of simulations
    iN = findall(N .≥ 1)  # positions in N such that N ≥ 1
    Y = zeros(0)
    for i ∈ iN
        Yi = rand(Pareto(2 + 1/N[i], δ), N[i])
        S[i] = sum(Yi)
        append!(Y, Yi)
    end
    println("P(S = 0) = ", 1 - sum(S .> 0)/m)
    ES, VS, MS, VaRS = mean(S), var(S), median(S), quantile(S, 0.995)
    return [S, ES, VS, MS, VaRS, N, Y]
end
```

Calculations about S

```
begin
    @time sim = simulateCRM(m = 1_000_000)
    println("E(S) = ", sim[2], "   V(S) = ", sim[3])
    println("M(S) = ", sim[4], "   VaR(0.995) = ", sim[5])
    println("range = [ ", minimum(sim[1]), " , ", maximum(sim[1]), " ]")
end
```

```
P(S = 0) = 0.874863
  0.101901 seconds (152.75 k allocations: 52.117 MiB, 13.78% gc time, 43.97%
compilation time)
E(S) = 0.3305663571543687   V(S) = 2.597828924148197
M(S) = 0.0   VaR(0.995) = 8.460361808120075
range = [ 0.0 , 477.4953922324707 ]
```




```
begin
    histogram(sim[1], label = "total claims", bins = 50)
    xaxis!("claims")
    yaxis!("frequency")
end
```

Calculations about $S \mid S > 0$

```
begin
    claims = sim[1][findall(sim[1] .> 0)]
    println("E(S | S > 0) = ", mean(claims), "   V(S | S > 0) = ", var(claims))
    println("M(S | S > 0) = ", median(claims), "   VaR(S | S > 0) = ",
    quantile(claims, 0.995))
    println("range = [ ", minimum(claims), " , ", maximum(claims), " ]")
end
```

```
E(S | S > 0) = 2.6416356245903985   V(S | S > 0) = 14.654972010065284
M(S | S > 0) = 1.5042940023123132   VaR(S | S > 0) = 21.642408203663354
range = [ 1.0000046967652276 , 477.4953922324707 ]
```




```
begin
    histogram(log.(claims), label = "log-claims", bins = 50)
    xaxis!("log-claims")
    yaxis!("log-frequency")
end
```